

316 **Заец Н. И.**

Радиолюбительские конструкции на PIC-микроконтроллерах. С алгоритмами работы программ и подробными комментариями к исходным текстам. Книга 2. — М.: СОЛОН-Пресс, 2010. — 192 с.: ил. — (Серия «СОЛОН — радиолюбителям»)

ISBN 5-98003-238-X

В книге даны новые примеры применения PIC-микроконтроллеров в радиолюбительской практике. Программисты найдут в книге программы с использованием встроенного в микроконтроллер модуля — АЦП и программы с различными внешними устройствами — термодатчиками типа DS18x20, LCD-дисплеями. Радиолюбители, которые желают повторить устройства, могут выбрать цифровой милливольтметр, для того чтобы защитить свой дом от перепадов напряжения, а трехфазный двигатель — от перегрузки. Термометр-часы, градусник и два терморегулятора будут полезными в любом доме. Ко всем программам даны алгоритмы работы и подробные комментарии.

Книга предназначена для широкого круга радиолюбителей, а также может быть полезна студентам, изучающим программирование микроконтроллеров.

УДК 621.31
ББК 32.96-04

Распространение ООО «Альянс-книга» (095) 258-91-94 www.abook.ru

www.solon-press.ru. E-mail: solon-avtor@coba.ru

КНИГА — ПОЧТОЙ

Книги издательства «СОЛОН-Пресс» можно заказать наложенным платежом (оплата при получении) по фиксированной цене. Заказ оформляется одним из двух способов:

1. Послать открытку или письмо по адресу: 123001, Москва, а/я 82.
2. Передать заказ по электронной почте на адрес: kniga@coba.ru.

Бесплатно высылается каталог издательства по почте.

При оформлении заказа следует правильно и полностью указать адрес, по которому должны быть высланы книги, а также фамилию, имя и отчество получателя. Желательно дополнительно указать свой телефон и адрес электронной почты.

Через Интернет вы можете в любое время получить свежий каталог издательства «СОЛОН-Пресс». Для этого надо послать пустое письмо на робот-автоответчик по адресу: katalog@solon-r.ru.

Устройства с использованием АЦП

Милливольтметр

В большинстве случаев для измерения напряжений и токов применяют электромеханические приборы (миллиамперметры) с установкой резистивного делителя либо шунта. Здесь описано применение трехразрядного цифрового милливольтметра, который можно использовать как вольтметр или амперметр постоянного тока в зависимости от того, где измеряется напряжение — на делителе или на шунте.

В лабораториях радиолюбителей есть много измерительной техники со стрелочными приборами. Как правило, точность показаний стрелочных приборов приходится проверять цифровыми приборами. Предлагаемое цифровое устройство можно использовать в любой измерительной технике с соответствующим преобразованием входного сигнала. По сути, предлагаемый милливольтметр является аналогом известной микросхемы 572ПВ2 с трехразрядной индикацией. Применение этого устройства может ограничить только ваша фантазия. Например, можно сделать малогабаритный щуп для измерения напряжений на исследуемой плате или щуп автомобилиста. Можно заменить стрелочные приборы в блоках питания или разработать измеритель емкости и индуктивности...

Основу прибора составляет дешевый микроконтроллер PIC16F676, который имеет 10-разрядный АЦП последовательного приближения. Источником опорного напряжения для АЦП служит напряжение питания микроконтроллера, поэтому точность измерения будет зависеть от стабильности питающего напряжения. При напряжении питания 5,12 В шаг квантования АЦП равен 5 мВ ($5,12/1024 = 0,005$). Если использовать делитель напряжения 1:20, то цена деления вольтметра будет равна 0,1 В, а максимальное измеряемое напряжение 51,2 В. Если установить делитель напряжения 1:2, то цена деления будет равна 0,01 В при максимальном измеряемом напряжении 5,12 В. При использовании шунта сопротивлением около 0,5 Ом цена деления амперметра составит 0,01 А. При этом если измеряемый ток составит 1 А, то падение напряжения на шунте 0,5 В. Очевидно, что для измерения больших токов необходимо устанавливать усилитель напряжения. Так, применив сопротивление шунта, равное 0,01 Ом, и усилитель напряжения, можно получить амперметр с пределом измерения до 10 А.

Для использования милливольтметра в различных приложениях положение запятой в разрядах программно не запитывается и может быть установлено по необходимости.

Алгоритм работы программы милливольтметра показан на рис. 1. Основная работа процессора — организация трехразрядной динамической индикации. Время индикации каждого разряда приблизительно равно 5 мс. Отсчет этого интервала времени организован с помощью прерывания по переполнению таймера TMR0. Таймер имеет коэффициент деления $K = 256$, предделитель $K = 4$ и счетчик $K = 5$. После индикации нулевого разряда выполняется измерение напряжения.

Результат измерения напряжения (10 разрядов) записывается в два регистра с правым выравниванием, т. е. младший регистр будет полностью заполнен, а в старшем регистре будут значащими только два младших разряда. Да-




Рис. 1. Алгоритм работы программы цифрового вольтметра

лее выполняется перекодировка 16 разрядов двоичного кода в три разряда двоично-десятичного кода. Результат перекодировки записывается в регистры индикации.

После индикации второго разряда программа повторяется.

На рис. 2 показана схема милливольтметра, который включен в режиме измерения напряжения до 51,2 В. Стабилизатор напряжения DA1 обеспечивает микроконтроллер напряжением питания и опорным напряжением АЦП. Для того чтобы использовать в резистивном делителе (1:20) резисторы R5, R6 с допустимыми отклонениями номиналов 5...10 %, напряжение питания можно изменять в небольших пределах подстроечным резистором R4.




Рис. 2. Милливольтметр. Измерение напряжения

Измерительный вход RA0 подключен к делителю R5, R6, и запитывается сегмент запятой Н в первом разряде индикатора. При измерении напряжения с пределом до 5,12 В коэффициент деления делителя R5, R6 необходимо установить равным 1:2, и запитать запятую во втором разряде.

Микроконтроллер DD1 тактируется внутренним генератором частотой около 4 МГц.

Включение милливольтметра в режиме измерения тока с пределом измерения, равным 5,12 А, показано на рис. 3. На операционном усилителе DA2.1 собран масштабирующий усилитель напряжения. Операционный усилитель DA2.2 включен по схеме повторителя. Если для вашей разработки необходим операционный усилитель, то можно использовать усилитель DA2.2, исключив повторитель из схемы милливольтметра. Для измерения тока до 10 А номинал резистора R13 должен иметь значение в пределах 27 кОм. В обоих случаях сегмент запятой Н запитывается во втором разряде.

Семисегментные светодиодные индикаторы можно заменить любыми с общим катодом, но тогда, возможно, потребуются подбор гасящих резисторов R3, R5—R11. Подстроечный резистор R4 типа СП5-16ВА можно заменить любым, не имеющим разрыва сопротивления при вращении движка резистора. Шунт R12 изготовлен из константанового или манганинового провода диаметром 1,5 мм, но можно использовать готовый от приборов типа М-830.




Рис. 3. Милливольтметр. Измерение тока

Налаживание вольтметра и амперметра начинают с установки напряжения питания 5,12 В вращением движка подстроечного резистора R4. При этом микроконтроллер не устанавливают в микросхемную панельку.

При налаживании вольтметра параллельно измеряемой цепи устанавливают цифровой вольтметр. Изменением напряжения питания микроконтроллера выравнивают показания настраиваемого прибора и цифрового вольтметра. При этом напряжение питания микроконтроллера не должно превышать максимально допустимое напряжение, равное 5,5 В. При уменьшении напряжения питания показания увеличиваются (уменьшается вес разряда квантования).

При налаживании амперметра последовательно с измеряемой цепью устанавливают цифровой амперметр. В отверстия печатной платы впаивают провод шунта с сопротивлением около 0,01 Ом. Постепенно уменьшают длину провода и добиваются максимального совпадения показаний настраиваемого прибора и цифрового амперметра. В качестве флюса для пайки шунта хорошо использовать ортофосфорную кислоту. Если слишком уменьшить длину шунта, то увеличение длины провода не даст результата, потому что часть провода будет покрыта припоем.

Точное совпадение показаний настраиваемого прибора и цифрового амперметра производят изменением напряжения питания микроконтроллера или сопротивления резистора R13. Если сопротивление шунта измерено с достаточной точностью, то подстройку выполняют только изменением напряжения питания.

Многие радиолюбители успели повторить конструкцию мультиметра по публикации статьи в журнале [1], но столкнулись с одной проблемой: при индикации восьмерки один или два младших разряда гаснут.

Автором проверена программа и схема мультиметра. Ошибок в них не обнаружено.

Выяснилось, что автор применял импортные светодиодные индикаторы, имеющие малые токи потребления. На схеме (рис. 2, 3) указаны их отечественные аналоги, имеющие большие токи потребления при такой же яркости. При включении всех сегментов (8!) общий вывод катода коротится и разряд выключается. Паспортные данные тока нагрузки на выход микроконтроллера 25 мА, но реально он оказался равным 15 мА. Возможно, это обусловлено тем, что в продаже трудно найти фирменные микроконтроллеры, имеющиеся же, как правило, азиатского происхождения. Простой выход из этого положения — это поставить резистор номиналом 5—10 Ом в цепь общих катодов младших разрядов. Резистор лучше подобрать по наилучшей яркости свечения, при которой восьмерка включится.

Печатная плата и расположение элементов для вольтметра показаны на рис. 4, 5, а для амперметра — на рис. 6, 7.




Рис. 4. Милливольтметр.
Печатная плата




Рис. 5. Милливольтметр. Расположение элементов




Рис. 6. Миллиамперметр.
Печатная плата




Рис. 7. Миллиамперметр.
Расположение элементов

Если микроконтроллер будет использоваться для других приложений, то необходимо считать и записать калибровочную константу. В данной работе калибровочная константа не учитывается, а регистр OSCAL установлен на минимальную частоту работы генератора.

```
; МИЛЛИВОЛЬТМЕТР.
; ИНДИКАЦИЯ — 7 СЕГМ. СВ. ДИОДЫ — 3 РАЗРЯДА.
; РАЗРАБОТАЛ НИКОЛАЙ ЗАЕЦ.
; ДЛЯ СВОБОДНОГО ИСПОЛЬЗОВАНИЯ РАДИОЛЮБИТЕЛЯМИ.
; п.ВЕЙДЕЛЕВКА, БЕЛГОРОДСКОЙ.
; saes@mail.ru
; ПРОГРАММА = MILLUM.ASM
; ВЕРСИЯ: 13-03-05.
; АССЕМБЛЕР И ОТЛАДЧИК: MPLAB IDE, ВЕРСИЯ: 5.70.40.
;

        #INCLUDE <P16F676.INC>
        __CONFIG 3FF4H
;=====
; ИСПОЛЬЗУЕТСЯ ВНУТРЕННИЙ ГЕНЕРАТОР ЧАСТОТОЙ 4,0 МГц.
;=====
; ПОРТ С — ВЫХОД СЕГМЕНТОВ А-Г.
; RA5 — ВЫХОД, СЕГМЕНТ G.
; RA1-RA2-RA4 — ВЫХОД РАЗРЯДОВ 0-2.
; RA0 — АНАЛОГОВЫЙ ВХОД 0.
;=====
; РЕГИСТРЫ РСН.
;=====
INDF      EQU 00H    ;ДОСТУП К ПАМЯТИ ЧЕРЕЗ FSR.
TIMER0    EQU 01H    ;TMR0.
OPTIONR    EQU 81H    ;OPTION (RP0=1).
PC         EQU 02H    ;СЧЕТЧИК КОМАНД.
STATUS     EQU 03H    ;РЕГИСТР СОСТОЯНИЯ АЛУ.
FSR        EQU 04H    ;РЕГИСТР КОСВЕННОЙ АДРЕСАЦИИ.
PORTA      EQU 05H    ;ПОРТ А ВВОДА/ВЫВОДА.
PORTC      EQU 07H    ;ПОРТ С ВВОДА/ВЫВОДА.
TRISA      EQU 85H    ;НАПРАВЛЕНИЯ ДАННЫХ ПОРТА А.
TRISC      EQU 87H    ;НАПРАВЛЕНИЯ ДАННЫХ ПОРТА С.
INTCON     EQU 0BH    ;РЕГИСТР ФЛАГОВ ПРЕРЫВАНИЙ.
PIR1       EQU 0CH    ;РАЗРЕШЕНИЯ ПЕРИФЕРИЙНЫХ ПРЕРЫВАНИЙ.
PIE1       EQU 8CH    ;РЕГИСТР РАЗРЕШЕНИЯ ПЕРИФЕРИЙНЫХ ПРЕРЫВАНИЙ.
T1CON      EQU 10H    ;РЕГИСТР УПРАВЛЕНИЯ ТАЙМЕРОМ 1.
CMCON      EQU 19H    ;РЕГИСТР УПРАВЛЕНИЯ КОМПАРАТОРОМ.
VRCON      EQU 99H    ;РЕГИСТР ИСТОЧНИКА ОПОРНОГО НАПРЯЖЕНИЯ.
PCON        EQU 8EH    ;КОНТРОЛЬ НАПРЯЖЕНИЯ.
ANSEL      EQU 91H    ;ВЫБОР АНАЛОГОВОГО ВХОДА.
WPUA       EQU 95H    ;ВКЛЮЧЕНИЯ ПОДТЯГИВАЮЩИХ РЕЗИСТОРОВ.
IOCA        EQU 96H    ;РАЗРЕШЕНИЕ ПРЕРЫВАНИЙ ПО ИЗМ. СИГН. НА ВХОДАХ А.
ADRESH     EQU 1EH    ;СТАРШИЙ РЕГИСТР АЦП.
ADRESL     EQU 9EH    ;МЛАДШИЙ РЕГИСТР АЦП.
ADCON0     EQU 1FH    ;УПРАВЛЕНИЯ АЦП.
ADCON1     EQU 9FH    ;ВЫБОР ТАКТИРОВАНИЯ АЦП.
;=====
; ОПРЕДЕЛЕНИЕ РОН.
;=====
TEKN EQU 20H    ;ДВОИЧНОЕ ИЗМЕРЕННОЕ ЗНАЧЕНИЕ НАПРЯЖЕНИЯ.
```

```

TEKL EQU 21H      ;
SEG EQU 22H       ;РЕГИСТР СЕГМЕНТОВ.
CEK EQU 23H       ;ВРЕМЯ ПАУЗЫ.
WTEMP EQU 24H     ;ВРЕМЕННЫЙ.
TEMP EQU 25H      ;ВРЕМЕННЫЙ.
FLAG EQU 26H      ;РЕГИСТР ФЛАГОВ.
COU EQU 27H       ;СЧЕТЧИК ПЕРЕКОДИРОВКИ.
STEMP EQU 28H     ;ВРЕМЕННЫЙ.
FTEMP EQU 29H     ;ВРЕМЕННЫЙ.
DEA EQU 2AH       ;РЕГИСТРЫ ПЕРЕКОДИРОВКИ.
EDA EQU 2BH       ;
EDI EQU 2CH       ;ПОРАЗРЯДНЫЕ РЕГИСТРЫ ИНДИКАЦИИ, 0.
DEI EQU 2DH       ;1.
COI EQU 2EH       ;2.
ANOD EQU 2FH      ;ПОЗИЦИОННОЕ ЗНАЧЕНИЕ КАТОДА (АНОДА).
;=====
; 1. ПУСК.
;=====
                ORG      0
                GOTO     INIT
                ORG      4
                GOTO     PRER
;=====
; 2. ТАБЛИЦА СЕГМЕНТОВ ДЛЯ ОБЩЕГО КАТОДА.
;=====
SEGDATA                ;7, 6, 5, 4, 3, 2, 1, 0.
                ADDWF    PCL,F      ;H, G, F, E, D, C, B, A.
; ОБЩИЙ              КАТОД              АНОД
RETLW           B'00111111'        ; 0 B'11000000'
RETLW           B'00000110'        ; 1 B'11111001'
RETLW           B'01011011'        ; 2 B'10100100'
RETLW           B'01001111'        ; 3 B'10110000'
RETLW           B'01100110'        ; 4 B'10011001'
RETLW           B'01101101'        ; 5 B'10010010'
RETLW           B'01111101'        ; 6 B'10000010'
RETLW           B'00000111'        ; 7 B'11111000'
RETLW           B'01111111'        ; 8 B'10000000'
RETLW           B'01101111'        ; 9 B'10010000'
;=====
; 3. ИНДИКАЦИЯ.
;=====
IND
                MOVLW     B'00010100' ;УСТАНОВЛИВАЕМ НОЛЬ В
                MOVWF     ANOD         ;0 РАЗЯДЕ ИНДИКАЦИИ.
                BCF       FLAG,1       ;СБРОСИМ ФЛАГ ПРЕРЫВАНИЯ.
                MOVWF     EDI          ;ЗАГРУЖАЕМ РЕГИСТР ИНДИКАЦИИ.
                CALL      INDZ         ;ПРОИНДИЦИРУЕМ.
                CALL      ADP0         ;ПРОВЕРИМ НАПРЯЖЕНИЕ.
                CALL      ZDEM         ;НА ОЖИДАНИЕ ПРЕРЫВАНИЯ.
IND1
                MOVLW     B'00010010' ;УСТАНОВЛИВАЕМ НОЛЬ В
                MOVWF     ANOD         ;1 РАЗЯДЕ.
                BCF       FLAG,1       ;ВСЕ ОСТАЛЬНЫЕ РАЗЯДЫ
                MOVWF     DEI          ;АНАЛОГИЧНО НУЛЕВОМУ.
                CALL      INDZ         ;
                CALL      ZDEM         ;НА ОЖИДАНИЕ ПРЕРЫВАНИЯ.

```

```

IND2
    MOVLW    B'00000110'    ;УСТАНОВЛИВАЕМ НОЛЬ В
    MOVWF    ANOD            ;ВО 2 РАЗРЯДЕ.
    BCF      FLAG,1          ;
    MOVFW    COI             ;
    CALL     INDZ            ;
    CALL     ZDEM            ;НА ОЖИДАНИЕ ПРЕРЫВАНИЯ.
    GOTO     IND             ;НА ИНДИКАЦИЮ НУЛЕВОГО РАЗРЯДА.

INDZ
    CALL     SEGDATA         ;ОПРЕДЕЛИМ СЕМИСЕКМЕНТНОЕ ЗНАЧЕНИЕ.
    MOVWF    SEG             ;ЗАПИШЕМ ЕГО В РЕГИСТР.
    MOVWF    PORTC           ;В ПОРТ С.
    MOVFW    ANOD            ;ВКЛЮЧИМ РАЗРЯД.
    MOVWF    PORTA           ;
    BTFSC    SEG,6           ;
    BSF      PORTA,5         ;
    RETURN                                ;ВЕРНЕМСЯ.

ZDEM
    BTFSC    FLAG,1          ;ОЖИДАНИЕ ПРЕРЫВАНИЯ.
    RETURN                                ;
    GOTO     ZDEM            ;

;=====
; 4. АЦП-ПРЕОБРАЗОВАНИЯ (ИЗМЕРЕНИЕ ВХОДНЫХ ВЕЛИЧИН).
;=====
ADP0
    MOVLW    B'10000001'    ;
    MOVWF    ADCON0          ;ВХОД 0, ВКЛЮЧЕНИЕ АЦП (УВХ).
    CALL     ZAD             ;
    BSF      ADCON0,1        ;ВКЛЮЧИМ ПРЕОБРАЗОВАНИЕ.
    BTFSC    ADCON0,1        ;ОЖИДАЕМ ЗАВЕРШЕНИЯ
    GOTO     $-1             ;ПРЕОБРАЗОВАНИЯ.
    MOVFW    ADRESH          ;ПЕРЕПИШЕМ РЕЗУЛЬТАТ ПРЕОБРАЗОВАНИЯ
    MOVWF    TEKH            ;В СТАРШИЙ ТЕКУЩИЙ РЕГИСТР.
    BSF      STATUS,5        ;ПЕРЕХОДИМ В БАНК 1.
    MOVLW    21              ;
    MOVWF    FSR             ;ПО КОСВЕННОЙ АДРЕСАЦИИ
    MOVFW    ADRESL          ;ЗАПИСЬ МЛ. РЕГИСТРА АЦП
    MOVWF    INDF            ;В РЕГИСТР ТЕКЛ.
    BCF      STATUS,5        ;ПЕРЕХОДИМ В БАНК 0.
    GOTO     BINDEC          ;ПЕРЕКОДИРУЕМ В 2_10 КОД.

ZAD
    MOVLW    .5              ;ЗАДЕРЖКА 20 мкс.
    ADDLW    -1              ;
    BTFSS    STATUS,2        ;
    GOTO     $-2             ;
    RETURN

;=====
; 5. ПЕРЕКОДИРОВКА ИЗ 16-РАЗРЯДНОГО 2-ГО В 5-РАЗРЯДНОЕ 2-10-Е.
; АЛГОРИТМ ПЕРЕКОДИРОВКИ ОСНОВЫВАЕТСЯ НА ПРИБАВЛЕНИИ 3 В МЛАДШИЙ И СТАРШИЙ
; ПОЛУБАЙТЫ. ЕСЛИ РЕЗУЛЬТАТ С ПЕРЕНОСОМ 1 В 3 РАЗРЯД (10=7+3),ТО
; ЗАПИСЫВАЕМ НОВОЕ ЗНАЧЕНИЕ В РЕГИСТР. ВЫПОЛНЯЕМ 16 РАЗ, СДВИГАЯ БИТЫ
; РЕГИСТРОВ.
;=====
BINDEC
    MOVLW    .16             ;ЗАПИШЕМ ЧИСЛО СДВИГОВ
    MOVWF    COU             ;В СЧЕТЧИК.

```

```

BIDE
    BCF      STATUS,0      ;ОБНУЛИМ БИТ "С".
    RLF      TEKL,1        ;СДВИНЕМ ПЕРЕКОДИРУЕМОЕ
    RLF      TEKH,1        ;ЧИСЛО, ПЕРЕМЕЩАЯ ЕГО СТАРШИЙ БИТ
    RLF      EDA,1         ;В МЛАДШИЙ БИТ РЕГИСТРОВ
    RLF      DEA,1         ;РЕЗУЛЬТАТА.
    DECFSZ   COU,1         ;ЗАФИКСИРУЕМ СДВИГ В СЧЕТЧИКЕ.
    GOTO     RASDEC        ;ПРОВЕРИМ ПОЛУБАЙТЫ НА СЕМЕРКУ.
    GOTO     MESTO         ;ЕСЛИ СЧЕТЧИК ПУСТ, ЗАПОЛНИМ РЕГИСТРЫ ИНДИКАЦИИ.

RASDEC
    MOVLW    EDA           ;ЗАПИШЕМ АДРЕС РЕГИСТРА
    MOVWF    FSR           ;В РЕГИСТР КОСВЕННОЙ АДРЕСАЦИИ.
    CALL     BCD           ;ПРОВЕРИМ ЗНАЧЕНИЕ РЕГИСТРА НА 7.
    MOVLW    DEA           ;АНАЛОГИЧНЫЕ ОПЕРАЦИИ ПРОДЕЛАЕМ
    MOVWF    FSR           ;С ДРУГИМИ РЕГИСТРАМИ.
    CALL     BCD           ;
    GOTO     BIDE          ;ПОЙДЕМ ПОВТОРЯТЬ СДВИГ.

BCD
    MOVLW    3             ;0000 0011
    ADDWF    0,0           ;ПРИБАВИМ 3 К РЕГИСТРУ И РЕЗУЛЬТАТ
    MOVWF    TEMP         ;ЗАПИШЕМ ВО ВРЕМЕННЫЙ РЕГИСТР.
    BTFSC    TEMP,3        ;ПРОВЕРИМ 3 БИТ, И ЕСЛИ ОН РАВЕН НУЛЮ,
    MOVWF    0             ;ПРОПУСКАЕМ ЗАПИСЬ РЕЗУЛЬТАТА В РЕГИСТР.
    MOVLW    30            ;48=0011 0000
    ADDWF    0,0           ;ПРИБАВИМ 3 К СТАРШЕМУ ПОЛУБАЙТУ РЕГИСТРА И
                           ;РЕЗУЛЬТАТ
    MOVWF    TEMP         ;ЗАПИШЕМ ВО ВРЕМЕННЫЙ РЕГИСТР.
    BTFSC    TEMP,7        ;ЕСЛИ БИТ ЕДИНИЧНЫЙ,
    MOVWF    0             ;ТО ЗАПИШЕМ НОВОЕ ЗНАЧЕНИЕ В РЕГИСТР.
    RETURN          ;ВЕРНЕМСЯ ДЛЯ ЗАГРУЗКИ НОВОГО ЗНАЧЕНИЯ РЕГИСТРА.
;=====
; 6. ИЗВЛЕКАЕМ ПОЛУБАЙТЫ ИЗ РЕГИСТРОВ СЧЕТА В РЕГИСТРЫ ИНДИКАЦИИ.
;=====
MESTO
    MOVLW    B'00001111';ИЗВЛЕКАЕМ ПОЛУБАЙТЫ
    ANDWF    DEA,0         ;В РЕГИСТРЫ ИНДИКАЦИИ.
    MOVWF    COI           ;
    MOVLW    B'11110000';
    ANDWF    EDA,0         ;
    MOVWF    DEI           ;
    SWAPF    DEI,1         ;
    MOVLW    B'00001111';
    ANDWF    EDA,0         ;
    MOVWF    EDI           ;
    CLRF     EDA           ;
    CLRF     DEA           ;
    RETURN          ;
;=====
; 7. СОХРАНЕНИЕ И ВОССТАНОВЛЕНИЕ ЗНАЧЕНИЙ РЕГИСТРОВ ПРИ ПРЕРЫВАНИИ.
;=====
PRER
    MOVWF    WTEMP         ;СОХРАНЕНИЕ ЗНАЧЕНИЙ РЕГИСТРОВ W И
    MOVWF    STATUS        ;STATUS,
    MOVWF    STEMP         ;
    MOVWF    FSR           ;FSR.
    MOVWF    FTEMP         ;

```

```

        BCF      STATUS, 6      ;
        BCF      STATUS, 5      ;
        INCF     CEK, 1         ;ДЕЛИТЕЛЬ НА 5.
        MOVLW    .5             ;
        SUBWF    CEK, 0         ;
        BTFSS    STATUS, 2      ;ЕСЛИ РАВНО 5,
        GOTO     REPER          ;
        BSF      FLAG, 1        ;ТО УСТАНОВИМ ФЛАГ 5 мс.
        CLRF     CEK            ;ОБНУЛИМ.
REPER    ;ВОССТАНОВЛЕНИЕ СОХРАНЕННЫХ ЗНАЧЕНИЙ.
        MOVWF    STEMP          ;ВОССТАНОВЛЕНИЕ РЕГИСТРОВ:
        MOVWF    STATUS         ;STATUS,
        MOVWF    FTEMP          ;
        MOVWF    FSR            ;FSR,
        MOVWF    WTEMP          ;W.
        BCF      INTCON, 2      ;СВРАСЫВАЕМ ФЛАГ ПРЕРЫВАНИЯ ОТ TMR0.
        RETFIE                 ;ВОЗВРАТ ИЗ ПРЕРЫВАНИЯ.
;=====
; 8. ИНИЦИАЛИЗАЦИЯ.
;=====
INIT
        BSF      STATUS, 5      ;ПЕРЕХОДИМ В БАНК 1.
        CLRF     OSCCAL^0x90    ;МИНИМАЛЬНАЯ ЧАСТОТА.
        MOVLW    0FFH          ;СИНХРОНИЗАЦИЯ ОТ RC.
        MOVWF    ADCON1         ;ТАКТ АЦП ОТ ВНУТРЕННЕГО ГЕНЕРАТОРА 500 кГц.
        MOVLW    B'10000001'    ;K=4.
        MOVWF    OPTION_REG^80H ;РЕЗИСТОРЫ ВЫКЛЮЧЕНЫ.
        MOVLW    B'10100000'    ;РАЗРЕШЕНИЕ ПРЕРЫВАНИЙ ОТ TMR0.
        MOVWF    INTCON         ;
        CLRF     PIE1^80H        ;ЗАПРЕЩЕНЫ ВСЕ ПЕРИФЕРИЙНЫЕ ПРЕРЫВАНИЯ.
        MOVLW    B'00000001'    ;ВСЕ ВЫХОДЫ. 0 — ВХОД АЦП.
        MOVWF    TRISA^80H      ;
        CLRF     TRISC^80H       ;ВСЕ ВЫХОДЫ.
        MOVLW    .1             ;
        MOVWF    ANSEL          ;ВЫБРАН АНАЛОГОВЫЙ ВХОД RA0/AN0.
        CLRF     VRCON          ;ИОН ОТКЛЮЧЕН.
        CLRF     PCON^80H        ;ПРЕРЫВАНИЯ ПО ПИТАНИЮ ЗАПРЕЩЕНЫ.
        CLRF     WPUA           ;ПОДТЯГИВАЮЩИЕ РЕЗИСТОРЫ ВЫКЛЮЧЕНЫ.
        CLRF     IOCA           ;ПРЕРЫВАНИЯ ЗАПРЕЩЕНЫ.
        BCF      STATUS, 5      ;ПЕРЕХОДИМ В БАНК 0.
        CLRF     PORTC          ;ВЫХОД И ИНДИКАТОРЫ ВЫКЛЮЧЕНЫ.
        CLRF     PORTA          ;
        CLRF     T1CON          ;ТАЙМЕР 1 ОТКЛЮЧЕН.
        MOVLW    .7             ;
        MOVWF    CMCON          ;КОМПАРАТОР ВЫКЛЮЧЕН.
        CLRF     FLAG           ;ВСЕ ОБНУЛЯЕМ И УСТАНОВЛИВАЕМ.
        CLRF     EDI            ;
        CLRF     DEI            ;
        CLRF     COI            ;
        CLRF     CEK            ;
        GOTO     IND
;=====
END
;=====

```

Цифровой прибор для блока питания с установкой защиты

Предлагаемый прибор необходим при налаживании устройств, особенно с помощью лабораторных блоков питания. Прибор защищает настраиваемое устройство от перегрузки по току и от превышения напряжения питания. Он также обеспечивает удобную цифровую индикацию тока и напряжения, установку пределов срабатывания защиты и их сохранение в энергонезависимой памяти.

В процессе проверки устройства возможно резкое увеличение потребляемого им тока из-за ошибок в монтаже или неправильных действий оператора. В результате могут выйти из строя дорогостоящие элементы. Для их защиты обычно применяют предохранители, среди которых предпочтительнее быстродействующие электронные, например [2, 3]. Но защита только по току недостаточна. Ошибка человека, управляющего лабораторным блоком питания, или пробой регулирующего транзистора в этом блоке вызовет повреждение налаживаемого устройства завышенным напряжением питания. Хотя защита по току обычно срабатывает и в этом случае, но уже после возникновения повреждений, так как именно они и вызывали ее срабатывание. Чаще всего эти повреждения необратимы. Защита по напряжению позволяет их предотвратить, поэтому она столь же необходима, как и по току.

Предлагаемый прибор предназначен для использования с блоком питания. Он обеспечивает цифровую индикацию напряжения и тока, потребляемого нагрузкой, раздельное включение и отключение защиты по току и напряжению, установку пределов срабатывания защиты. Результат измерений отображается двумя четырехразрядными индикаторами. На время отключения прибора установки защиты сохраняются в энергонезависимой памяти.

Прибор разработан на основе 28-выводного микроконтроллера PIC16F873, имеющего 10-разрядный АЦП.

Основные технические характеристики

Пределы измерения напряжения, В	0...50
Пределы измерения тока, А	0...9,99
Пределы срабатывания защиты:	
по току, А	от 0,01 до 9,99 с шагом 0,01
по напряжению, В	от 0,1 до 50 с шагом 0,1
Время срабатывания защиты:	
среднее при одной включенной защите, мс.	0,075
среднее, при двух включенных защитах, мс	0,15
максимальное, мс	1
Напряжение питания прибора, В	9...40
Максимальный потребляемый ток, мА	50

В случае необходимости верхний предел измерения тока и срабатывания защиты может быть увеличен изменением номиналов делителя и шунта. По совокупности характеристик прибор превосходит ранее описанные защитные

устройства аналогичного назначения [4, 5] (как самостоятельные, так и встроенные в блоки питания), что удалось достичь в результате использования микроконтроллера.

Алгоритм работы программы цифрового прибора показан на рис. 8. После инициализации регистров микроконтроллера из памяти выбираются ранее установленные значения защиты по току и напряжению. Все операции по измерению входных величин и их перекодировка выполняются процессором в паузах между выводом информации на восьмиразрядный индикатор. Время индикации одного разряда определяется временем, которое занимает 10 циклов измерения входного напряжения и тока. Поскольку измерение входных величин во время установки не выполняется, а во время вывода на индикацию к времени измерения прибавляется время перекодировки, то становится понятным, что время индикации каждого разряда — величина переменная. Младшие четыре разряда индикатора предназначены для вывода значения тока.

После индикации нулевого разряда выполняются 10 циклов измерения тока и напряжения. Фактически микроконтроллер измеряет напряжение, но для простоты мы будем говорить об измерении тока. Цикл одного измерения занимает примерно 75 мкс, поэтому время реагирования защиты на отклонение от установленного значения будет равно 150 мкс. Во время вывода на индикацию (1 раз через 0,5 с) время реагирования увеличивается до 1 мс.

Поскольку алгоритм работы программ измерения одинаков, то на рис. 8 показан только один из них. После вывода на индикацию каждого разряда устанавливается счетчик циклов измерения. Измерение входных величин начинается с выбора входа и включения АЦП. Считываются старший и младший регистры измерения с правым выравниванием, т. е. старшие шесть разрядов равны нулю. Считанные значения сравниваются на превышение установленных величин. Если измеренное значение больше установленного, то выключается управляющий выход и нагрузка отключается. Вся последующая программа продолжает выполняться, но повторно включить выход можно только после перезапуска микроконтроллера. Если результат сравнения отрицательный, то проверяется флаг прерывания. Если флаг прерывания установлен и прошло 0,5 с, то измеренное значение перекодируется в двоично-десятичный код и результат переписывается в регистры индикации. Следующий цикл индикации будет выполняться с новыми значениями в регистрах индикации.

Для исключения мигания индикаторов при граничных показаниях вывод на индикацию производится через 0,5 с. Счетчик 0,5 с организован при помощи 8-разрядного таймера TMR0 и 8 разрядного предделителя, включенного перед таймером. При переполнении таймера выполняется прерывание, заполняется счетчик прерываний и после восьмого прерывания устанавливается флаг 0,5 с.

Если 0,5 с не прошло, то декрементируется счетчик циклов измерения и проверяется на ноль. Если значение счетчика не равно нулю, то цикл измерения повторяется. Если счетчик пуст, то выполняется индикация следующего разряда.

После индикации седьмого разряда проверяется состояние кнопок «Установка» и «Разряд». Если производится установка, то на единицу увеличивается число в выбранном разряде. После любого изменения значения защиты вы-




Рис. 8. Алгоритм работы программы цифрового прибора для блока питания с установкой защиты

полняется перекодировка двоично-десятиричного числа в двоичное число. Двоичное число необходимо для выполнения быстрого сравнения измеренного значения и установленного. Установленные двоично-десятиричное и двоичное числа записываются в энергонезависимую память. Если нажата кнопка «Разряд», то прибавляется единица в регистр запятой и программа переходит к выполнению циклов измерения. Во время индикации разряда с числом, равным числу, установленному в регистре запятой, в разряде включится запятая.

Далее циклы индикации и измерения повторяются.

На рис. 9 показана принципиальная схема прибора. Напряжение питания прибора должно быть в пределах 9...40 В (предпочтительнее использовать интервал 9...12 В). Нагрузку (налаживаемое устройство) подключают к выходу прибора. На вход прибора подают напряжение питания нагрузки, которое не должно превышать 50 В. Общие провода прибора и обоих источников питания соединены между собой и образуют единый общий провод. Однако с ним не должен быть соединен общий провод нагрузки, так как переключательный транзистор VT1 включен в разрыв минусового провода питания нагрузки. Допустимо питать прибор от того же источника, что и нагрузку. При этом плюсовые провода входа и питания должны быть соединены. Однако нагрузка в этом случае защищена хуже. Предпочтительнее питать прибор от отдельного источника.

Порты В и С микроконтроллера DD1 выводят информацию на светодиодные индикаторы HG1, HG2 с общим катодом. Входы RA4, RA5 задействованы кнопками «Установка» и «Разряд». Источником опорного напряжения АЦП при измерении служит напряжение питания микроконтроллера. Порт RA2 запрограммирован как выход, управляющий полевым переключательным транзистором VT1. RA0 и RA1 — входы АЦП для измерения напряжения и тока соответственно. Стабилизатор на микросхеме DA1 вырабатывает напряжение питания 5,12 В для микроконтроллера DD1 и операционного усилителя DA2. В небольших пределах напряжение питания можно изменять подстроечным резистором R6, что используется при калибровке прибора.

Датчик напряжения — резисторный делитель R8R9, напряжение с которого подается непосредственно на вход RA0 микроконтроллера. Датчик тока — резистор R20. Напряжение на нем усиливается ОУ DA2.1 с коэффициентом, примерно равным 48. Далее оно через повторитель на ОУ DA2.2 поступает на вход RA1 микроконтроллера.

Топология печатной платы и расположение элементов на ней представлены на рис. 10, 11. Печатная плата для блока индикаторов типа CC56—12GWA показана на рис. 12.

Микроконтроллер PIC16F873 без изменения рисунка печатной платы можно заменить микроконтроллером PIC16F876. Резистор R20 взят из сгоревшего прибора типа М-830—М-838. Его можно изготовить из константанового или манганинового провода диаметром 1,5 мм. Если прибор не будет работать в условиях с большим перепадом температур (гараж—комната), то резистор можно сделать из нихрома. Для диаметра провода 1,5 мм сопротивлением 0,01 Ом длина провода из нихрома составит приблизительно 18 мм [6]. Добиться точного значения сопротивления провода нет необходимости, так как проще подобрать сопротивление резистора R11 при калибровке измерителя тока. Операционный усилитель типа LM358 можно взять с любой буквой или его аналог КР1040УД1.

Семисегментные светодиодные индикаторы можно заменить любыми с общим катодом, но тогда, возможно, потребуются подбор гасящих резисторов R12—R19. Подстроечный резистор R6 типа СП5-16ВА можно взять любым, не имеющим разрыва сопротивления при вращении движка резистора.

Полевой транзистор лучше взять с пониженным напряжением затвора (с буквой «L») из таблицы, приведенной в статье [7].




Рис. 9. Цифровой прибор для блока питания с установкой защиты

Налаживание прибора начинают с установки напряжения питания на выводе 2 стабилизатора DA1. Вращением движка подстроечного резистора R6 устанавливают напряжение, равное 5,12 В. На время установки напряжения микроконтроллер извлекают из микросхемной панельки. Затем устанавливают запрограммированный микроконтроллер и подключают цифровой измеритель-




Рис. 10. Топология печатной платы




Рис. 11. Расположение элементов на плате

льный прибор (можно типа М-830) на измерение напряжения с пределом в младшем разряде 0,1 В. Сравнивают показания измерительного прибора и старших четырех разрядов настраиваемого прибора. Совпадения показаний добиваются изменением в небольших пределах напряжения питания микроконтроллера резистором R6. При этом напряжение питания микроконтроллера не должно превышать максимально допустимое напряжение, равное 5,5 В.

Налаживание измерителя тока начинают с установки вместо постоянного резистора R11 переменного номиналом 51 кОм. Последовательно с нагрузкой подключают цифровой амперметр с пределом в младшем разряде 10 мА. Если

Содержание

К читателям	3
Устройства с использованием АЦП	5
Милливольтметр	5
Цифровой прибор для блока питания с установкой защиты	15
Автомат защиты от перепадов сетевого напряжения	37
Устройство защиты от перепадов сетевого напряжения	66
Устройство защиты без индикации	86
Устройство защиты трехфазных двигателей	93
Устройства, измеряющие температуру	108
Термометр-часы.....	108
Градусник.....	135
Два терморегулятора.....	148
Приложение.	
Коды прошивок микроконтроллеров	174
Литература	187

Готовится к выпуску!

Заец Н. И.

Радиолюбительские конструкции на PIC-микроконтроллерах. Книга 3

Третья книга расширяет диапазон применения PIC-микроконтроллеров в радиолюбительской практике. В книге дан пример программы с использованием встроенного в микроконтроллер модуля USART и различных внешних устройств — LCD-дисплеев и ЖКИ, выполненных по COG-технологии. Радиолюбители, которые желают повторить устройства могут выбрать: охрану подворья, шахматные часы, таймеры на 7 и 9 выходов, а также автомат кормления аквариумных рыб. Для родной школы можно изготовить простое устройство подачи звонков по расписанию.

В отдельную главу вынесены «трудные темы» взаимодействия микроконтроллеров с внешними устройствами: ЖК-дисплеями и термодатчиками типа DS18x20.

Ко всем программам даны алгоритмы работы и подробные комментарии.

К книге прилагается **КОМПАКТ-ДИСК**, содержащий 48 исходных текстов программ ко всем устройствам четырех книг автора, («Электронные самоделки. Для быта, отдыха и здоровья» и «Радиолюбительские конструкции на PIC-микроконтроллерах» в трех книгах), вышедших в издательстве СОЛОН-Пресс (www.solon-press.ru, E-mail: solon-avtor@coba.ru) справочные материалы по микроконтроллерам на русском и английском языках, установочные программы для программаторов и ассемблера MPASM.

Книга предназначена для широкого круга радиолюбителей, а также может быть полезна студентам, изучающим программирование микроконтроллеров.